
1. はじめに
現状のところ，この文書はVer.0 です． 単に古典命題論理の強完全性定理 (定理 53) の証明を Lean

で形式化しただけのものであり，解説や導入が非常に不十分です． 一応ほとんどの事実に対して証明

をつけていますが，ある程度既に知っている人向けの内容になっています． 実験的に，普通の数式表

示 (e.g. 𝜑) とソースコード上の表示 (e.g. φ) を混在させて書いていますが，読みづらいかもしれませ

ん． ぜひフィードバックをいただければ幸いです．

2. 論理式の導入
この節は Logic101/CPL/Formula.lean に置いてある．

定義 1 (Formula) :  型 α の要素を命題変数とした論理式 Formula α を次のように定義する．

1. Formula.atom: 型 α の要素 a は論理式である．論理式であることは #a と書く．

2. Formula.bot: 論理定項 ⊥ は論理式である．

3. Formula.and: φ と ψ が論理式であれば，φ ⋏ ψ も論理式である．

4. Formula.or: φ と ψ が論理式であれば，φ ⋎ ψ も論理式である．

5. Formula.imp: φ と ψ が論理式であれば，φ → ψ も論理式である．

6 inductive Formula (α : Type u)
7 | atom : α → Formula α
8 | bot : Formula α
9 | and : Formula α → Formula α → Formula α
10 | or  : Formula α → Formula α → Formula α
11 | imp : Formula α → Formula α → Formula α
12 deriving DecidableEq

16 prefix:90 "#" => Formula.atom
17 notation:max "⊥" => Formula.bot
18 infixr:85 " ➝ " => Formula.imp
19 infixl:84 " ⋏ " => Formula.and

20 infixl:83 " ⋎ " => Formula.or

便宜上，∼φ を φ → ⊥ と定義する．また，⊤ を ∼⊥ と定義する．

22 @[grind] def neg (φ : Formula α) : Formula α := φ ➝ ⊥
23 prefix:90 "∼" => Formula.neg
24
25 @[grind] def top : Formula α := ∼⊥
26 notation:max "⊤" => Formula.top

命題変数が列挙可能なら論理式は適当に列挙可能である． この事実は完全性定理の証明に必要となる

が，その証明は普通の数学上では自明なものであるし，コード上でも技術的に面倒なだけでつまらな

いため，証明は省略する．

補題 2 (Formula.instEncodable) :  α が Encodable であるなら Formula α も Encodable である． 

すなわち，次を満たす．

1. Formula α から自然数への全域関数 encode : Formula α → ℕ が存在する．

2. 自然数から Formula α への部分関数 decode : ℕ ⇀ Option (Formula α) が存在する．

3. 任意の φ : Formula α について，decode (encode φ) は存在し，かつその値は φ と等しい．

34 def toNat : Formula α → ℕ

41 def ofNat : ℕ → Option (Formula α)

84 lemma ofNat_toNat : ∀ (φ : Formula α), ofNat (toNat φ) = some φ

89 instance : Encodable (Formula α) where
90   encode := toNat
91   decode := ofNat
92   encodek := ofNat_toNat

定義 3 (Theory) :  Formula α の集合を理論 Theory α と呼ぶ．

98 abbrev Theory (α) := Set (Formula α)

3. 証明体系の導入
証明体系として Hilbert流を採用する． やや冗長な量の公理を導入しているが，これは今後直観主義

命題論理などを扱うことを企画してのことである．

「証明」と「証明できる」ことを区別していることに注意する．前者はオブジェクトであり，後者は

命題である．

定義 4 (Proof) :  Formula α 上の証明 Proof φ を帰納的に定義する．Proof φ を ⊢! φ と書く．

1. 次の論理式たちは公理であり，それ単体で証明になる．

• andElimL: φ ⋏ ψ → φ

• andElimR: φ ⋏ ψ → ψ

• andIntro: φ → ψ → (φ ⋏ ψ)

• orIntroL: φ → (φ ⋎ ψ)

• orIntroR: ψ → (φ ⋎ ψ)

• orElim: (φ → χ) → (ψ → χ) → (φ ⋎ ψ → χ)

• impS: (φ → ψ → χ) → (φ → ψ) → (φ → χ)

• impK: φ → ψ → φ

• efq: ⊥ → φ

• lem: φ ⋎ ∼φ

2. mdp: φ → ψ の証明 と φ の証明から，ψ の証明が得られる．

5 inductive Proof : Formula α → Type*
6 | andIntro! {φ ψ} : Proof $ φ ➝ ψ ➝ (φ ⋏ ψ)
7 | andElimL! {φ ψ} : Proof $ (φ ⋏ ψ) ➝ φ
8 | andElimR! {φ ψ} : Proof $ (φ ⋏ ψ) ➝ ψ
9 | orIntroL! {φ ψ} : Proof $ φ ➝ (φ ⋎ ψ)
10 | orIntroR! {φ ψ} : Proof $ ψ ➝ (φ ⋎ ψ)
11 | orElim! {φ ψ χ} : Proof $ (φ ⋎ ψ) ➝ (φ ➝ χ) ➝ (ψ ➝ χ) ➝ χ
12 | impS! {φ ψ χ} : Proof $ (φ ➝ (ψ ➝ χ)) ➝ (φ ➝ ψ) ➝ (φ ➝ χ)
13 | impK!   {φ ψ} : Proof $ φ ➝ ψ ➝ φ
14 | mdp!      {φ ψ} : Proof (φ ➝ ψ) → Proof φ → Proof ψ
15 | efq!        {φ} : Proof $ ⊥ ➝ φ
16 | lem!        {φ} : Proof $ φ ⋎ ∼φ
17 prefix:40 "⊢! " => Proof

定義 5 (Provable, Unprovable) :  φ の証明 ⊢! φ が存在するとき φ は証明可能 Provable φ である

と言い，⊢ φ と書く．

19 def Provable (φ : Formula α) := Nonempty (⊢! φ)
20 prefix:40 "⊢ " => Provable

証明可能ではないとき，φ は証明不能 Unprovable φ であると言い，⊬ φ と書く．

22 abbrev Unprovable (φ : Formula α) := ¬(⊢ φ)
23 prefix:40 "⊬ " => Unprovable

明らかにわかることだが，証明不能であることは証明が存在しないことと同値である．

以降では証明という概念を陽には扱わず，証明可能性で議論することにする．

補題 6 (Provable.impId) : ⊢ φ ➝ φ

証明 :

1. φ ➝ (φ ➝ φ) ➝ φ は impK より証明可能．

2. (φ ➝ (φ ➝ φ) ➝ φ) ➝ (φ ➝ φ ➝ φ) ➝ φ ➝ φ は impS より証明可能．

3. 1. と 2. より mdp から (φ ➝ φ ➝ φ) ➝ φ ➝ φ は証明可能．

4. φ ➝ φ ➝ φ は impK より証明可能．

5. 3. と 4. より mdp から φ ➝ φ が証明可能．

62 lemma impId : ⊢ φ ➝ φ := by
63   have h₁ : ⊢ φ ➝ (φ ➝ φ) ➝ φ := impK;
64   have h₂ : ⊢ (φ ➝ (φ ➝ φ) ➝ φ) ➝ (φ ➝ φ ➝ φ) ➝ φ ➝ φ := impS;
65   have h₃ : ⊢ (φ ➝ φ ➝ φ) ➝ φ ➝ φ := mdp h₂ h₁;
66   have h₄ : ⊢ φ ➝ φ ➝ φ := impK;
67   exact mdp h₃ h₄;

❏

証明可能という概念は技術的に扱いにくいので，もうすこし扱いやすくするために演繹という概念を

導入する．

定義 7 (Deduction) :  Theory α である Γ から φ への演繹 Deduction Γ φ を帰納的に定義する．

Deduction Γ φ を Γ ⊢! φ と書く．

1. φ ∈ Γ ならば，Γ ⊢! φ を得て良い．

2. 証明 ⊢! φ から Γ ⊢! φ を得て良い．

3. Γ ⊢! φ ➝ ψ と Γ ⊢! φ から，Γ ⊢! ψ を得て良い．

75 | of_context! {Γ φ} : φ ∈ Γ → Deduction Γ φ
76 | of_proof!  {Γ φ} : ⊢! φ → Deduction Γ φ
77 | mdp! {Γ φ ψ} : Deduction Γ (φ ➝ ψ) → Deduction Γ φ → Deduction Γ ψ
78 infix:40 " ⊢! " => Deduction

定義 8 (Deducible) :  Γ から φ への演繹 Γ ⊢! φ が存在するとき，φ は Γ から演繹可能 Deducible 

Γ φ であると言い，Γ ⊢ φ と書く．

81 infix:40 " ⊢ " => Deducible

そうでないとき，φ は Γ から演繹不能 Underiavlbe Γ φ であると言い，Γ ⊬ φ と書く．

84 infix:40 " ⊬ " => Undeducible

補題 9 (Deducible.weakening) :  Γ ⊆ Δ かつ Γ ⊢ φ ならば，Δ ⊢ φ

証明 : 演繹可能性の帰納法による．

(of_context) φ ∈ Γ ならば仮定を使えば φ ∈ Δ であるから，Δ ⊢! φ である．

(of_proof) ⊢! φ ならば Δ ⊢! φ である．

(mdp) Γ ⊢! φ ➝ ψ と Γ ⊢! φ ならば，帰納法の仮定から Δ ⊢! φ ➝ ψ と Δ ⊢! φ であるから，mdp 

を使えば Δ ⊢! ψ である．

189 lemma weakening (hΓΔ : Γ ⊆ Δ) (h : Γ ⊢ φ) : Δ ⊢ φ := by
190   induction h using Deducible.rec with
191   | of_context h =>
192     apply of_context;
193     exact hΓΔ h;
194   | of_provable h =>

195     apply of_provable;
196     exact h;
197   | mdp d₁ d₂ ih₁ ih₂ =>
198     apply mdp;
199     . apply ih₁ hΓΔ;
200     . apply ih₂ hΓΔ;

❏

補題 9 の系として直ちに次が得られる．

系 10 (Deducible.weakening_insert) :  Γ ⊢ φ ならば，insert ψ Γ ⊢ φ

205 @[grind =]
206 theorem deduction_theorem : (insert φ Γ) ⊢ ψ ↔ Γ ⊢ (φ ➝ ψ) := by

演繹という概念を導入したのは演繹定理と呼ばれる便利な 定理 11 を証明するためである．

定理 11 (演繹定理，Deducible.deduction_theorem) :  insert φ Γ ⊢ ψ と Γ ⊢ φ ➝ ψ は同値であ

る．

証明 :

(⟹) insert φ Γ ⊢ ψ 上の帰納法．

(of_context) ψ ∈ insert φ Γ ならば場合分けする．

• ψ = φ のとき，Γ ⊢ φ ➝ φ であり，impId より成り立つ．

• ψ ∈ Γ のとき，Γ ⊢ ψ であるから，afortiori より Γ ⊢ φ ➝ ψ である．

(of_provable) ⊢ ψ ならば証明可能性上の afortiori より ⊢ φ ➝ ψ であり，of_provable より 

Γ ⊢ φ ➝ ψ である．

(mdp) 帰納法の仮定より Γ ⊢ φ ➝ ψ ➝ χ と Γ ⊢ φ ➝ ψ であるから，impS より Γ ⊢ φ ➝ χ であ

る．

(⟸) Γ ⊢ φ ➝ ψ なら 系 10 より insert φ Γ ⊢ φ ➝ ψ である． これと仮定から mdp を使えば 

insert φ Γ ⊢ ψ である．

206 theorem deduction_theorem : (insert φ Γ) ⊢ ψ ↔ Γ ⊢ (φ ➝ ψ) := by
207   constructor;
208   . intro h;
209     generalize eΔ : insert φ Γ = Δ at h;
210     induction h using Deducible.rec with
211     | of_context h =>
212       subst eΔ;
213       simp at h;
214       rcases h with (rfl | h);
215       . exact impId;
216       . apply afortiori
217         apply of_context h;
218     | of_provable h =>
219       apply of_provable;

220       exact Provable.afortiori h;
221     | mdp _ _ ih₁ ih₂ =>
222       subst eΔ;
223       replace ih₁ := ih₁ (by grind);
224       replace ih₂ := ih₂ (by grind);
225       apply ruleImpS ih₁ ih₂;
226   . intro h;
227     apply mdp (weakening_insert h);
228     apply of_context;
229     simp;

❏

補題 12 (Deducible.ruleImpTrans) :  Γ ⊢ φ ➝ ψ および Γ ⊢ ψ ➝ χ ならば，Γ ⊢ φ ➝ χ

証明 :

1. 定理 11 より仮定 Γ ⊢ φ ➝ ψ は insert φ Γ ⊢ ψ と同値である．

2. もう片方の仮定 Γ ⊢ ψ ➝ χ は 系 10 より insert φ Γ ⊢ ψ ➝ χ である．

3. 1. と 2. より mdp を使えば insert φ Γ ⊢ χ である．

4. 再び 定理 11 より Γ ⊢ φ ➝ χ である．

231 lemma ruleImpTrans : Γ ⊢ φ ➝ ψ → Γ ⊢ ψ ➝ χ → Γ ⊢ φ ➝ χ := by
232   intro h₁ h₂;
233   replace h₁ : insert φ Γ ⊢ ψ := deduction_theorem.mpr h₁;
234   replace h₂ : insert φ Γ ⊢ ψ ➝ χ := weakening_insert h₂;
235   exact deduction_theorem.mp $ mdp h₂ h₁;

❏

補題 13 (Deducible.ruleEFQNeg) :  Γ ⊢ ∼φ ならば，Γ ⊢ φ ➝ ψ

証明 : ∼φ が φ ➝ ⊥ であることに注意すれば efq と 補題 12 より直ちに従う．

237 lemma ruleEFQNeg : Γ ⊢ ∼φ → Γ ⊢ φ ➝ ψ := λ h => ruleImpTrans h efq

❏

補題 14 (Deducible.ruleDNE) :  Γ ⊢ ∼∼φ ならば，Γ ⊢ φ

証明 : ruleOrElim と lem より Γ ⊢ φ ➝ φ と Γ ⊢ ∼φ ➝ φ を言えば良い．前者は impId から，後者

は Γ ⊢ ∼∼φ と 補題 13 より従う．

239 lemma ruleDNE (h : Γ ⊢ ∼∼φ) : Γ ⊢ φ := by
240   apply ruleOrElim lem;
241   . show Γ ⊢ φ ➝ φ;
242     exact impId;
243   . apply ruleEFQNeg h;

❏

4. 意味論の導入
この節は Logic101/CPL/Semantics.lean に置いてある．

定義 15 (Valuation) :  写像 α → Prop を付値 Valuation と呼ぶ．

6 abbrev Valuation (α) := α → Prop

定義 16 (Forces) :  論理式 φ : Formula α が付値 V : Valuation α によって強制される Forces V 

φ という関係を次のように定義する．Forces V φ を V ⊩ φ と書く．

1. V ⊩ #a は付値の値 V a である．

2. V ⊩ ⊥ は成立しない．

3. V ⊩ φ ⋏ ψ は V ⊩ φ かつ V ⊩ ψ のとき成立する．

4. V ⊩ φ ⋎ ψ は V ⊩ φ または V ⊩ ψ のとき成立する．

5. V ⊩ φ → ψ は V ⊩ φ ならば V ⊩ ψ のとき成立する．

9 def Forces (V : Valuation α) : Formula α → Prop
10 | #a        => V a
11 | ⊥         => False
12 | φ ⋏ ψ     => Forces V φ ∧ Forces V ψ
13 | φ ⋎ ψ     => Forces V φ ∨ Forces V ψ
14 | φ ➝ ψ     => Forces V φ → Forces V ψ
15 infix:50 " ⊩ " => Forces

∼φ は φ → ⊥ であり，⊤ は ∼⊥ であることから，定義から次が成り立つ．

補題 17 :

• V ⊩ ∼φ は V ⊩ φ でないときに成立する．

• V ⊩ ⊤ は常に成立する．

19 @[grind .] lemma forces_top : V ⊩ ⊤ := by grind;
20 @[grind =] lemma forces_neg : V ⊩ ∼φ ↔ ¬(V ⊩ φ) := by grind;

定義 18 (Valid) :  論理式 φ が妥当 Valid φ であるとは，任意の付値 V に対して V ⊩ φ が成立す

ることである．Valid φ を ⊨ φ と書く． あるいは φ がトートロジー Tautology であるとも言う．

24 def Valid (φ : Formula α) : Prop := ∀ V, V ⊩ φ
25 prefix:50 "⊨ " => Valid
26 alias Tautology := Valid

定義 19 (ConsequenceOn) :  付値 V と理論 Γ と論理式 φ について，∀ ψ ∈ Γ, V ⊩ ψ ならば V ⊩ φ 

が成立するとき， V において Γ が φ から帰結する ConsequenceOn V Γ φ という．ConsequenceOn 

V Γ φ を Γ ⊨[V] φ と書く．

28 @[grind .]

29
def ConsequenceOn (V : Valuation α) (Γ : Theory α) (φ : Formula α) : Prop := (∀ ψ ∈ 

Γ, V ⊩ ψ) → V ⊩ φ
30 notation Γ " ⊨[" V "] " φ => ConsequenceOn V Γ φ

定義 20 (Consequence) :  任意の付値 V について ConsequenceOn V Γ φ が成立するとき，Γ が φ か

ら帰結する Consequence Γ φ という．Consequence Γ φ を Γ ⊨ φ と書く．

34 def Consequence (Γ : Theory α) (φ : Formula α) : Prop := ∀ V, Γ ⊨[V] φ
35 infix:50 " ⊨ " => Consequence

論理式が妥当であることと空な理論から意味論的に帰結することは同値である．

補題 21 (iff_valid_forall_forces) :  ⊨ φ は ∅ ⊨ φ と同値である．

38
lemma iff_valid_forall_forces : ⊨ φ ↔ ∅ ⊨ φ := by simp [Valid, Consequence, 

ConsequenceOn]

定理 22 (健全性，sound) :  Γ ⊢ φ ならば Γ ⊨ φ

証明 : 演繹可能性の帰納法による．

(of_context) 任意の付値 V を取り， ∀ ψ ∈ Γ, V ⊩ ψ を仮定する． このとき φ ∈ Γ だから V ⊩ 

φ が成り立ち，V の任意性より Γ ⊨ φ が成り立つ．

(andIntroL など) 各公理などに対して V ⊩ φ ⋏ ψ → φ や V ⊩ φ ➝ ψ ➝ φ などが成り立つことを

確認すれば良い．

(mdp) 任意の付値 V と ∀ ψ ∈ Γ, V ⊩ ψ を仮定する． 帰納法の仮定より Γ ⊨ φ ➝ ψ と Γ ⊨ φ が

成り立つから，特に V ⊩ φ ➝ ψ と V ⊩ φ が成り立つ． よって V ⊩ ψ が成り立つ．

40 theorem sound : Γ ⊢ φ → Γ ⊨ φ := by
41   rintro h;
42   induction h using Deducible.rec_provable with
43   | orElim =>
44     intro V hV;
45     grind [Forces];
46   | mdp _ _ ihφψ ihφ =>
47     intro V hV;
48     exact ihφψ V hV $ ihφ V hV;
49   | _ => grind;

❏

系 23 (sound_provable) :  ⊢ φ ならば ⊨ φ

51 lemma sound_provable : ⊢ φ → ⊨ φ := by

証明 : 定理 22 で Γ として ∅ を取り，補題 21 を用いれば良い． ❏

系 24 (consistent) :  ⊬ ⊥．すなわち，どのような論理式でも証明可能である訳では無いという

意味で証明体系は矛盾していない。

証明 : さもなくば 系 23 から ⊨ ⊥ である．適当な付値 V を取れば V ⊩ ⊥ となり，定義に反する．

56 theorem consistent : ⊬ (⊥ : Formula α) := by
57   by_contra h;
58   apply sound_provable h (λ _ => True);

❏

健全性の逆方向である完全性を示すためには様々な準備が必要である．

定義 25 (Theory.Inconsistent, Theory.Consistent) :  理論 Γ が矛盾している 

Theory.Inconsistent Γ とは Γ ⊢ ⊥ であることを指す． そうでないとき Γ は無矛盾 

Theory.Consistent Γ であると言う．

69 abbrev Inconsistent (Γ : Theory α) : Prop := Γ ⊢ ⊥
70
71 abbrev Consistent (Γ : Theory α) : Prop := ¬(Γ.Inconsistent)

補題 26 (Theory.iff_inconsistent_insert， Theory.iff_consistent_insert) :

1. insert φ Γ が無矛盾であることは Γ ⊢ ∼φ と同値である．

2. 逆に insert ∼φ Γ が矛盾していることは Γ ⊬ ∼φ と同値である．

証明 : 演繹定理(定理 11)の主張そのものである．

73
@[grind =] lemma iff_inconsistent_insert : (insert φ Γ).Inconsistent ↔ Γ ⊢ ∼φ := 

deduction_theorem
74

75
@[grind =] lemma iff_consistent_insert : (insert φ Γ).Consistent ↔ Γ ⊬ ∼φ := 

deduction_theorem.not

❏

補題 27 (Theory.either_unprovable_of_consistent) :  Γ が無矛盾であるならば，Γ ⊬ φ または Γ 

⊬ ∼φ のいずれかが成り立つ．

証明 : もしそうでないとすると Γ ⊢ φ かつ Γ ⊢ ∼φ である． したがって mdp から Γ ⊢ ⊥ であり，

Γ は無矛盾であるという仮定に反する．

87
lemma either_unprovable_of_consistent (Γ_consis : Γ.Consistent) (φ) : Γ ⊬ φ ∨ Γ ⊬ 

∼φ := by
88   contrapose! Γ_consis;
89   obtain ⟨hφ, hnφ⟩ := Γ_consis;
90   apply mdp hnφ hφ;

❏

補題 28 :  Γ ⊬ φ ならば，insert ∼φ Γ は無矛盾である．

証明 : 対偶を示す．insert ∼φ Γ が矛盾しているとすると，補題 26 より Γ ⊢ ∼∼φ であり，

ruleDNE より Γ ⊢ φ である．

105
lemma consistent_insert_neg_of_unprovable (h : Γ ⊬ φ) : (insert (∼φ) 

Γ).Consistent := by
106   contrapose! h;
107   apply ruleDNE;
108   apply iff_inconsistent_insert.mp h;

❏

定義 29 (Maximal) :  理論 Γ が極大 Maximal Γ であるとは，任意の論理式 φ について φ ∈ Γ また

は ∼φ ∈ Γ が成り立つことである．

110 abbrev Maximal (Γ : Theory α) : Prop := ∀ φ, φ ∈ Γ ∨ ∼φ ∈ Γ

定義 30 (MaximalConsistentTheory) :  無矛盾かつ極大な理論 Γ : Theory α を極大無矛盾理論 

MaximalConsistentTheory α と呼ぶ．

115 abbrev MaximalConsistentTheory (α) := { Γ : Theory α // Γ.Consistent ∧ Γ.Maximal }

次の 定理 31 は Lindenbaum補題と呼ばれる重要な定理であり，完全性定理の証明において中核を成

す． この定理の証明には Leanにおいてメタのレベルで排中律を用いていることに注意されたい．

定理 31 (lindenbaum) :  Γ が無矛盾であるならば，ある極大無矛盾理論 Δ が存在して Γ ⊆ Δ を満

たす．

220
lemma lindenbaum (Γ : Theory α) (Γ_consis : Theory.Consistent Γ) : ∃ Δ : 

MaximalConsistentTheory α, Γ ⊆ Δ.1 := by

定理 31 の証明にはいくつかの補助的な定義や補題が必要である．

定義 32 (lindenbaum_expand) :  理論 Γ と論理式 φ について，lindenbaum_expand Γ φ を次のよう

に定義する．

• insert φ Γ が無矛盾であるならば，lindenbaum_expand Γ φ = insert φ Γ

• そうでなければ，lindenbaum_expand Γ φ = Γ

122 open Classical in

123
def lindenbaum_expand (Γ : Theory α) (φ : Formula α) : Theory α := if 

Theory.Consistent (insert φ Γ) then insert φ Γ else Γ

注意 33 :  ここで，insert φ Γ が無矛盾であるかは現段階では決定不能（Decidableではない）

であるが，Classical を用いて決定可能としていることに注意せよ．

補題 34 (consistent_lindenbaum_expand) :  Γ が無矛盾であるならば，lindenbaum_expand Γ φ も

無矛盾である．

証明 : 定義及び仮定から直ちに従う．

125 @[grind =>]

126
lemma consistent_lindenbaum_expand (Γ_consis : Theory.Consistent Γ) : 

Theory.Consistent (lindenbaum_expand Γ φ) := by
127   dsimp [lindenbaum_expand];
128   split <;> assumption;

❏

以下 α は列挙可能だとすると，補題 2 から論理式は列挙可能であるとしてよい．

定義 35 (lindenbaum_indexed) :  自然数で添字付けられた理論の族 lindenbaum_indexed : ℕ → 

Theory α を次のように定義する．

• lindenbaum_indexed 0 = Γ

• lindenbaum_indexed (n + 1) はもし n に対応する論理式 φ が存在するとき lindenbaum_expand 

(lindenbaum_indexed n) φ とし，存在しないときは lindenbaum_indexed n とする．

132 def lindenbaum_indexed (Γ : Theory α) : ℕ → Theory α
133   | 0     => Γ
134   | n + 1 =>
135     match (decode n) with
136     | some φ => lindenbaum_expand (lindenbaum_indexed Γ n) φ
137     | none   => lindenbaum_indexed Γ n

補題 36 (consistent_lindenbaum_indexed) :  Γ が無矛盾であるならば，任意の自然数 n について 

lindenbaum_indexed n も無矛盾である．

証明 : 仮定および 補題 34 から n の帰納法によって示される．

142
lemma consistent_lindenbaum_indexed (Γ_consis : Theory.Consistent Γ) : 

Theory.Consistent (lindenbaum_indexed Γ n) := by
143   induction n with
144   | zero => assumption;
145   | succ n ih =>
146     dsimp [lindenbaum_indexed];
147     split;
148     . grind [consistent_lindenbaum_expand];
149     . assumption;

❏

補題 37 (monotone_lindenbaum_indexed) :  任意の自然数 n ≤ m について，lindenbaum_indexed n 

⊆ lindenbaum_indexed m が成り立つ．

証明 : n = m のとき自明．n ≤ m で成立すると仮定して n ≤ m + 1 の場合を示す． 明らかに定義

から lindenbaum_indexed m ⊆ lindenbaum_indexed (m + 1) であるから ⊆ の推移律と帰納法の仮

定から従う．

151
lemma monotonic_lindenbaum_indexed (h : n ≤ m) : lindenbaum_indexed Γ n ⊆ 

lindenbaum_indexed Γ m := by
152   induction h with
153   | refl => simp;
154   | step h ih =>
155     trans;

156     . exact ih;
157     . dsimp [lindenbaum_indexed, lindenbaum_expand];
158       split;
159       . split <;> grind;
160       . grind;

❏

定義 38 (lindenbaum_maximal) :  理論 lindenbaum_maximal をすべての自然数 n における 

lindenbaum_indexed n の和集合として定義する．

162 def lindenbaum_maximal (Γ : Theory α) : Theory α := ⋃ n, lindenbaum_indexed Γ n

補題 39 :  φ ∈ Γ.lindenbaum_maximal はある自然数 n が存在して φ ∈ lindenbaum_indexed n であ

ることと同値である．

証明 : 定義より明らか．

164
lemma iff_mem_lindenbaum_maximal : φ ∈ Γ.lindenbaum_maximal ↔ ∃ n, φ ∈ 

Γ.lindenbaum_indexed n := by
165   simp [lindenbaum_maximal, Set.mem_iUnion]

❏

補題 40 (subset_lindenbaum_indexed_lindenbaum_maximal, subset_lindenbaum_maximal) :

1. 任意の自然数 n について，lindenbaum_indexed n ⊆ Γ.lindenbaum_maximal が成り立つ．

2. 特に Γ ⊆ Γ.lindenbaum_maximal が成り立つ．

証明 : 定義より明らか．

164
lemma iff_mem_lindenbaum_maximal : φ ∈ Γ.lindenbaum_maximal ↔ ∃ n, φ ∈ 

Γ.lindenbaum_indexed n := by
165   simp [lindenbaum_maximal, Set.mem_iUnion]
166
167 @[grind .]

168
lemma subset_lindenbaum_indexed_lindenbaum_maximal : lindenbaum_indexed Γ n ⊆ 

lindenbaum_maximal Γ := by
169   intro φ hφ;
170   apply iff_mem_lindenbaum_maximal.mpr;
171   use n;
172

173
lemma subset_lindenbaum_maximal : Γ ⊆ lindenbaum_maximal Γ := 

subset_lindenbaum_indexed_lindenbaum_maximal (n := 0)

❏

補題 41 (exists_deducible_maximal) :  Γ.lindenbaum_maximal ⊢ φ ならば，ある自然数 n が存在

して lindenbaum_indexed n ⊢ φ．

証明 : Γ.lindenbaum_maximal ⊢ φ に対して演繹可能性についての帰納法を行う．

12



(of_context) φ ∈ Γ.lindenbaum_maximal ならば，定義からある自然数 n が存在して φ ∈ 

lindenbaum_indexed n である． よって lindenbaum_indexed n ⊢ φ である．

(of_proof) 証明可能であるならそもそも文脈無しで演繹可能だから，n = 0 とすれば良い．

(mdp) Γ.lindenbaum_maximal ⊢ φ ➝ ψ と Γ.lindenbaum_maximal ⊢ φ に対して 帰納法の仮定より

ある n と m に対して lindenbaum_indexed n ⊢ φ ➝ ψ と lindenbaum_indexed m ⊢ φ が成り立つ． 

このとき所望の自然数は max n m とすればよい． なぜなら n ≤ max n m かつ m ≤ max n m で

あるから，補題 37 より lindenbaum_indexed (max n m) ⊢ φ ➝ ψ と lindenbaum_indexed (max 

n m) ⊢ φ が成り立ち，mdp より lindenbaum_indexed (max n m) ⊢ ψ が成り立つ．

175
lemma exists_deducible_maximal (h : Γ.lindenbaum_maximal ⊢ φ) : ∃ n, 

lindenbaum_indexed Γ n ⊢ φ := by
176   generalize eΔ : Γ.lindenbaum_maximal = Δ at h;
177   induction h using Deducible.rec with
178   | of_context h =>
179     subst eΔ;
180     obtain ⟨n, hn⟩ := iff_mem_lindenbaum_maximal.mp h;
181     use n;
182     apply of_context hn;
183   | of_provable h =>
184     use 0;
185     exact of_provable h;
186   | mdp d₁ d₂ ih₁ ih₂ =>
187     obtain ⟨n, ih₁⟩ := ih₁ eΔ;
188     obtain ⟨m, ih₂⟩ := ih₂ eΔ;
189     use max n m;
190     apply mdp;
191     . exact weakening (monotonic_lindenbaum_indexed (by omega)) ih₁;
192     . exact weakening (monotonic_lindenbaum_indexed (by omega)) ih₂;

❏

補題 42 (consistent_lindenbaum_maximal) :  Γ が無矛盾なら Γ.lindenbaum_maximal も無矛盾であ

る．

証明 : もしそうでないとすると，補題 41 よりある自然数 n が存在して lindenbaum_indexed n ⊢ 

⊥ であるが，それは 補題 36 に反する．

194 @[grind .]

195
lemma consistent_lindenbaum_maximal (Γ_consis : Theory.Consistent Γ) : 

Theory.Consistent (Γ.lindenbaum_maximal) := by
196   by_contra hC;
197   obtain ⟨_, hn⟩ := exists_deducible_maximal hC;
198   apply consistent_lindenbaum_indexed Γ_consis hn;

❏

補題 43 (maximal_lindenbaum_maximal) :  Γ が無矛盾なら Γ.lindenbaum_maximal は極大である．

証明 : 任意に論理式 φ を取る． 補題 42 より Γ.lindenbaum_maximal は無矛盾だから， 補題 27 

より Γ.lindenbaum_maximal ⊬ φ または Γ.lindenbaum_maximal ⊬ ∼φ が成り立つ．

Γ.lindenbaum_maximal ⊬ φ の場合を考える． このとき insert (∼φ) (Γ.lindenbaum_indexed 

(encode (∼φ))) は無矛盾でないと仮定すると Γ.lindenbaum_indexed (encode (∼φ)) ⊢ ∼∼φ である

から，ruleDNE より Γ.lindenbaum_indexed (encode (∼φ)) ⊢ φ が成り立つ．したがって 補題 9 と 

補題 41 より Γ.lindenbaum_maximal ⊢ φ が成り立ち，仮定に反する． したがって insert (∼φ) 

(Γ.lindenbaum_indexed (encode (∼φ))) は無矛盾であるから，∼φ ∈ Γ.lindenbaum_indexed 

(encode (∼φ) + 1) が成り立ち，補題 39 より ∼φ ∈ Γ.lindenbaum_maximal が成り立つ．

後者の場合も同様にして φ ∈ Γ.lindenbaum_maximal が成り立つ．

いずれにせよ φ ∈ Γ.lindenbaum_maximal または ∼φ ∈ Γ.lindenbaum_maximal が成り立つから 

Γ.lindenbaum_maximal は極大である．

200
lemma maximal_lindenbaum_maximal (Γ_consis : Theory.Consistent Γ) : Theory.Maximal 

(Γ.lindenbaum_maximal) := by
201   intro φ;

202
  rcases Γ.lindenbaum_maximal.either_unprovable_of_consistent 

(consistent_lindenbaum_maximal Γ_consis) φ with (hC | hC);
203   . right;
204     apply iff_mem_lindenbaum_maximal.mpr;
205     use ((encode (∼φ)) + 1);
206     suffices (insert (∼φ) (Γ.lindenbaum_indexed (encode (∼φ)))).Consistent by
207       simp [lindenbaum_indexed, lindenbaum_expand, this];
208     apply iff_consistent_insert.mpr;
209     contrapose! hC;
210     apply weakening (by grind) (ruleDNE hC);
211   . left;
212     apply iff_mem_lindenbaum_maximal.mpr;
213     use ((encode φ) + 1);
214     suffices (insert φ (Γ.lindenbaum_indexed (encode φ))).Consistent by
215       simp [lindenbaum_indexed, lindenbaum_expand, this]
216     apply iff_consistent_insert.mpr;
217     contrapose! hC;
218     apply weakening (by grind) hC;

❏

これで 定理 31 の証明の準備が整った．

証明 (定理 31) : 無矛盾な Γ を取り，その Γ.lindenbaum_maximal を Δ とすれば，Γ ⊆ Δ であるこ

とは 補題 40 より従い， Δ が無矛盾であることは 補題 42 より従い， Δ が極大であることは 補

題 43 より従う．

220
lemma lindenbaum (Γ : Theory α) (Γ_consis : Theory.Consistent Γ) : ∃ Δ : 

MaximalConsistentTheory α, Γ ⊆ Δ.1 := by

221
  use ⟨Γ.lindenbaum_maximal, consistent_lindenbaum_maximal Γ_consis, 

maximal_lindenbaum_maximal Γ_consis⟩;
222   exact subset_lindenbaum_maximal;

❏

注意 44 :  実際には，定理 31 はMathlibにある Zornの補題といくつかの集合に関する補題を

用意すればこれよりも簡潔に証明できる．

次は極大無矛盾理論に対して性質を見ていく．以下 Γ は極大無矛盾理論であるとする．

補題 45 (MaximalConsistentTheory.iff_mem_provable) :  論理式 φ について，φ ∈ Γ は Γ ⊢ φ と同

値である．

証明 :

(⟹) of_context から明らか．

(⟸) Maximal であるから ∼φ ∉ Γ であることを言えば良い． もし ∼φ ∈ Γ ならば Γ ⊢ ∼φ である

から仮定 Γ ⊢ φ と mdp より Γ ⊢ ⊥ が言えて，Γ が無矛盾であることに反する．

237   constructor;
238   . apply of_context;
239   . intro h;
240     suffices ∼φ ∉ Γ.1 by grind;
241     by_contra hC;
242     apply Γ.consistent;
243     apply mdp;
244     . show Γ.1 ⊢ ∼φ;
245       apply of_context hC;
246     . exact h;

❏

補題 46 (MaximalConsistentTheory.not_mem_bot) :  ⊥ ∉ Γ．

証明 : 補題 45 と Γ の無矛盾性から直ちに従う．

248 @[simp, grind .] lemma not_mem_bot : ⊥ ∉ Γ.1 := by grind;

❏

補題 47 (MaximalConsistentTheory.iff_mem_neg) :  ∼φ ∈ Γ は φ ∉ Γ と同値である．

証明 : 補題 27 と 補題 45 より直ちに従う．

250 @[grind =] lemma iff_mem_neg : ∼φ ∈ Γ.1 ↔ φ ∉ Γ.1 := by constructor <;> grind;

❏

補題 48 (MaximalConsistentTheory.iff_not_mem_provable) :  φ ∉ Γ は Γ ⊢ ∼φ と同値である．

証明 : 補題 47 と 補題 45 より直ちに従う．

252 @[grind =] lemma iff_not_mem_provable : φ ∉ Γ.1 ↔ Γ.1 ⊢ ∼φ := by grind;

❏

補題 49 (MaximalConsistentTheory.iff_mem_imp) :  φ ➝ ψ ∈ Γ は「φ ∈ Γ ならば ψ ∈ Γ」と同値で

ある．

証明 :

(⟹) 補題 45 より Γ ⊢ φ ➝ ψ，Γ ⊢ φ ならば Γ ⊢ ψ であることを証明すれば良い． これは mdp 

そのものである．

(⟸) 「φ ∈ Γ ならば ψ ∈ Γ」 を仮定することは「φ ∉ Γ または ψ ∈ Γ」のどちらかと同値であ

る．

φ ∉ Γ と仮定すると 補題 47 より ∼φ ∈ Γ であるから，Γ ⊢ ∼φ であり，ruleEFQNeg より Γ ⊢ φ 

➝ ψ が成り立つ．

ψ ∈ Γ と仮定すると Γ ⊢ ψ であるから，afortiori より Γ ⊢ φ ➝ ψ が成り立つ．

いずれにせよ Γ ⊢ φ ➝ ψ が成り立つ．

255 lemma iff_mem_imp : (φ ➝ ψ) ∈ Γ.1 ↔ (φ ∈ Γ.1 → ψ ∈ Γ.1) := by
256   constructor;
257   . simp only [iff_mem_provable]
258     apply mdp;
259   . rintro h;
260     rcases not_or_of_imp h with (hφ | hψ);
261     . replace hφ := iff_mem_neg.mpr hφ;
262       simp only [iff_mem_provable] at hφ ⊢;
263       exact ruleEFQNeg hφ;
264     . simp only [iff_mem_provable] at hψ ⊢;
265       exact afortiori hψ;

❏

注意 50 :  ⟸ において「 𝐴 ならば 𝐵 」を「 𝐴 でない，または 𝐵 」の同値だと見做すこと

(not_or_of_imp)は Lean上のメタ論理として古典論理でしか成立しない．

補題 51 (MaximalConsistentTheory.iff_mem_and) :  φ ⋏ ψ ∈ Γ は φ ∈ Γ かつ ψ ∈ Γ と同値である．

証明 : 補題 45 より Γ ⊢ φ ⋏ ψ は Γ ⊢ φ かつ Γ ⊢ ψ と同値であることを示せば良い． ⟹ 方向は 

ruleAndElimL と ruleAndElimR から，⟸ 方向は ruleAndIntro から従う．

268 lemma iff_mem_and : (φ ⋏ ψ) ∈ Γ.1 ↔ φ ∈ Γ.1 ∧ ψ ∈ Γ.1 := by
269   simp only [iff_mem_provable]
270   constructor;
271   . intro h;
272     constructor;
273     . apply ruleAndElimL h;
274     . apply ruleAndElimR h;
275   . simp only [and_imp];
276     apply ruleAndIntro;

❏

補題 52 (MaximalConsistentTheory.iff_mem_or) :  φ ⋎ ψ ∈ Γ は φ ∈ Γ または ψ ∈ Γ と同値であ

る．

証明 :

(⟹) φ ⋎ ψ ∈ Γ なので Γ ⊢ φ ⋎ ψ である． φ ∉ Γ かつ ψ ∉ Γ だと仮定すると，補題 48 より Γ 

⊢ ∼φ かつ Γ ⊢ ∼ψ である． したがって ruleOrElim より Γ ⊢ ⊥ であり，Γ が無矛盾であること

に反する． よって φ ∈ Γ または ψ ∈ Γ．

(⟸) 補題 45 より Γ ⊢ φ または Γ ⊢ ψ ならば Γ ⊢ φ ⋎ ψ であることを示せば良い． これは 

ruleOrIntroL または ruleOrIntroR から従う．

279 lemma iff_mem_or : (φ ⋎ ψ) ∈ Γ.1 ↔ φ ∈ Γ.1 ∨ ψ ∈ Γ.1 := by
280   constructor;
281   . intro h;
282     by_contra! hC;
283     obtain ⟨hφ, hψ⟩ := hC;
284     replace hφ : ↑Γ ⊢ ∼φ := iff_not_mem_provable.mp hφ;
285     replace hψ : ↑Γ ⊢ ∼ψ := iff_not_mem_provable.mp hψ;
286     apply Γ.consistent;
287     apply ruleOrElim (iff_mem_provable.mp h) hφ hψ;
288   . simp only [iff_mem_provable];
289     rintro (hφ | hψ);
290     . apply ruleOrIntroL;
291       exact hφ;
292     . apply ruleOrIntroR;
293       exact hψ;

❏

これで完全性定理を示す全ての準備が整った．

定理 53 (完全性，complete) :  Γ ⊨ φ ならば Γ ⊢ φ

証明 : 対偶を示す． Γ ⊬ φ と仮定すると，補題 28 より insert (∼φ) Γ は無矛盾である．した

がって 定理 31 より極大無矛盾理論 Δ が存在して insert (∼φ) Γ ⊆ Δ を満たすものが取れる．

このとき付値 V を #a が Δ に属するとき成立すると定める． すると，論理式の帰納法および 補

題 46， 補題 49，補題 51，補題 52 より V ⊩ ψ は ψ ∈ Δ と同値であることが示せる．

したがって，Γ ⊆ Δ であるから ∀ ψ ∈ Γ, V ⊩ ψ が成り立つ．また，∼φ ∈ Δ であるから V ⊩ ∼φ が

成り立つ．すなわち V ⊩ φ でない．

よって Γ ⊨ φ は成り立たない．

297 theorem complete [Encodable α] : Γ ⊨ φ → Γ ⊢ φ := by
298   contrapose!;
299   intro h;
300   suffices ∃ V, (∀ ψ ∈ Γ, V ⊩ ψ) ∧ ¬V ⊩ φ by simpa [Consequence, ConsequenceOn];
301

302
  obtain ⟨Δ, hΓΔ⟩ := (insert (∼φ) Γ).lindenbaum $ by grind 

[Theory.consistent_insert_neg_of_unprovable];
303
304   let V : Valuation α := λ a => (#a ∈ Δ.1);
305   have hV : ∀ {ψ}, V ⊩ ψ ↔ ψ ∈ Δ.1 := by
306     intro ψ;
307     induction ψ <;> grind
308   use V;
309   constructor;
310   . intro γ hγ;
311     apply hV.mpr;
312     apply hΓΔ;
313     grind;
314   . apply hV.not.mpr;

315     apply Δ.iff_mem_neg.mp;
316     apply hΓΔ;
317     grind;

❏

5. コード全文
1 import Mathlib.Logic.Encodable.Basic
2
3 variable {α : Type u}
4
5
6 inductive Formula (α : Type u)
7 | atom : α → Formula α
8 | bot : Formula α
9 | and : Formula α → Formula α → Formula α
10 | or  : Formula α → Formula α → Formula α
11 | imp : Formula α → Formula α → Formula α
12 deriving DecidableEq
13
14
15 namespace Formula
16
17 prefix:90 "#" => Formula.atom
18 notation:max "⊥" => Formula.bot
19 infixr:85 " ➝ " => Formula.imp
20 infixl:84 " ⋏ " => Formula.and
21 infixl:83 " ⋎ " => Formula.or
22
23 @[grind] def neg (φ : Formula α) : Formula α := φ ➝ ⊥
24 prefix:90 "∼" => Formula.neg
25
26 @[grind] def top : Formula α := ∼⊥
27 notation:max "⊤" => Formula.top
28
29 section
30
31 variable [Encodable α]
32 open Encodable
33
34 def toNat : Formula α → ℕ
35   | ⊥     => (Nat.pair 0 0) + 1
36   | #a  => (Nat.pair 1 <| encode a) + 1
37   | φ ➝ ψ   => (Nat.pair 2 <| φ.toNat.pair ψ.toNat) + 1
38   | φ ⋏ ψ   => (Nat.pair 3 <| φ.toNat.pair ψ.toNat) + 1
39   | φ ⋎ ψ   => (Nat.pair 4 <| φ.toNat.pair ψ.toNat) + 1
40
41 def ofNat : ℕ → Option (Formula α)
42   | 0 => none
43   | e + 1 =>
44     let idx := e.unpair.1
45     let c := e.unpair.2
46     match idx with
47     | 0 => some ⊥
48     | 1 => (decode c).map (#·)
49     | 2 =>
50       have : c.unpair.1 < e + 1 := Nat.lt_succ_iff.mpr
51         $ le_trans (Nat.unpair_left_le _)
52         $ Nat.unpair_right_le _
53       have : c.unpair.2 < e + 1 := Nat.lt_succ_iff.mpr
54         $ le_trans (Nat.unpair_right_le _)
55         $ Nat.unpair_right_le _
56       do
57         let φ <- ofNat c.unpair.1
58         let ψ <- ofNat c.unpair.2
59         return φ ➝ ψ

60     | 3 =>
61       have : c.unpair.1 < e + 1 := Nat.lt_succ_iff.mpr
62         $ le_trans (Nat.unpair_left_le _)
63         $ Nat.unpair_right_le _
64       have : c.unpair.2 < e + 1 := Nat.lt_succ_iff.mpr
65         $ le_trans (Nat.unpair_right_le _)
66         $ Nat.unpair_right_le _
67       do
68         let φ <- ofNat c.unpair.1
69         let ψ <- ofNat c.unpair.2
70         return φ ⋏ ψ
71     | 4 =>
72       have : c.unpair.1 < e + 1 := Nat.lt_succ_iff.mpr
73         $ le_trans (Nat.unpair_left_le _)
74         $ Nat.unpair_right_le _
75       have : c.unpair.2 < e + 1 := Nat.lt_succ_iff.mpr
76         $ le_trans (Nat.unpair_right_le _)
77         $ Nat.unpair_right_le _
78       do
79         let φ <- ofNat c.unpair.1
80         let ψ <- ofNat c.unpair.2
81         return φ ⋎ ψ
82     | _ => none
83
84 lemma ofNat_toNat : ∀ (φ : Formula α), ofNat (toNat φ) = some φ
85   | #a => by simp [toNat, ofNat, Nat.unpair_pair, encodek];
86   | ⊥  => by simp [toNat, ofNat]
87   | φ ➝ ψ | φ ⋏ ψ | φ ⋎ ψ   => by simp [toNat, ofNat, ofNat_toNat φ, ofNat_toNat ψ]
88
89 instance : Encodable (Formula α) where
90   encode := toNat
91   decode := ofNat
92   encodek := ofNat_toNat
93
94 end
95
96 end Formula
97
98
99 abbrev Theory (α) := Set (Formula α)

1 import Logic101.CPL.Formula
2
3 variable {α} {φ ψ χ : Formula α} {Γ Δ : Theory α}
4
5 inductive Proof : Formula α → Type*
6 | andIntro! {φ ψ} : Proof $ φ ➝ ψ ➝ (φ ⋏ ψ)
7 | andElimL! {φ ψ} : Proof $ (φ ⋏ ψ) ➝ φ
8 | andElimR! {φ ψ} : Proof $ (φ ⋏ ψ) ➝ ψ
9 | orIntroL! {φ ψ} : Proof $ φ ➝ (φ ⋎ ψ)

10 | orIntroR! {φ ψ} : Proof $ ψ ➝ (φ ⋎ ψ)
11 | orElim! {φ ψ χ} : Proof $ (φ ⋎ ψ) ➝ (φ ➝ χ) ➝ (ψ ➝ χ) ➝ χ
12 | impS! {φ ψ χ} : Proof $ (φ ➝ (ψ ➝ χ)) ➝ (φ ➝ ψ) ➝ (φ ➝ χ)
13 | impK!   {φ ψ} : Proof $ φ ➝ ψ ➝ φ
14 | mdp!      {φ ψ} : Proof (φ ➝ ψ) → Proof φ → Proof ψ
15 | efq!        {φ} : Proof $ ⊥ ➝ φ
16 | lem!        {φ} : Proof $ φ ⋎ ∼φ
17 prefix:40 "⊢! " => Proof
18
19 def Provable (φ : Formula α) := Nonempty (⊢! φ)
20 prefix:40 "⊢ " => Provable
21
22 abbrev Unprovable (φ : Formula α) := ¬(⊢ φ)
23 prefix:40 "⊬ " => Unprovable
24
25 namespace Provable
26
27 open Proof
28

29 @[grind .] lemma andElimL : ⊢ (φ ⋏ ψ) ➝ φ := ⟨andElimL!⟩
30 @[grind .] lemma andElimR : ⊢ (φ ⋏ ψ) ➝ ψ := ⟨andElimR!⟩
31 @[grind .] lemma andIntro : ⊢ φ ➝ ψ ➝ (φ ⋏ ψ) := ⟨andIntro!⟩
32 @[grind .] lemma orIntroL : ⊢ φ ➝ (φ ⋎ ψ) := ⟨orIntroL!⟩
33 @[grind .] lemma orIntroR : ⊢ ψ ➝ (φ ⋎ ψ) := ⟨orIntroR!⟩
34 @[grind .] lemma orElim : ⊢ (φ ⋎ ψ) ➝ (φ ➝ χ) ➝ (ψ ➝ χ) ➝ χ := ⟨orElim!⟩
35 @[grind .] lemma impS : ⊢ (φ ➝ (ψ ➝ χ)) ➝ (φ ➝ ψ) ➝ (φ ➝ χ) := ⟨impS!⟩
36 @[grind .] lemma impK : ⊢ φ ➝ ψ ➝ φ := ⟨impK!⟩
37 @[grind .] lemma efq : ⊢ ⊥ ➝ φ := ⟨efq!⟩
38 @[grind .] lemma lem : ⊢ φ ⋎ ∼φ := ⟨lem!⟩
39 @[grind! →] lemma mdp : ⊢ φ ➝ ψ → ⊢ φ → ⊢ ψ := λ ⟨d₁⟩ ⟨d₂⟩ => ⟨mdp! d₁ d₂⟩
40
41 @[induction_eliminator]
42 lemma rec
43   {motive   : (φ : Formula α) → (⊢ φ) → Sort}
44   (andElimL : ∀ {φ ψ}, motive ((φ ⋏ ψ) ➝ φ) andElimL)
45   (andElimR : ∀ {φ ψ}, motive ((φ ⋏ ψ) ➝ ψ) andElimR)
46   (andIntro  : ∀ {φ ψ}, motive (φ ➝ ψ ➝ (φ ⋏ ψ)) andIntro)
47   (orIntroL  : ∀ {φ ψ}, motive (φ ➝ (φ ⋎ ψ)) orIntroL)
48   (orIntroR  : ∀ {φ ψ}, motive (ψ ➝ (φ ⋎ ψ)) orIntroR)
49   (orElim    : ∀ {φ ψ χ}, motive ((φ ⋎ ψ) ➝ (φ ➝ χ) ➝ (ψ ➝ χ) ➝ χ) orElim)
50   (impS    : ∀ {φ ψ χ}, motive ((φ ➝ (ψ ➝ χ)) ➝ (φ ➝ ψ) ➝ (φ ➝ χ)) impS)
51   (impK    : ∀ {φ ψ}, motive (φ ➝ ψ ➝ φ) impK)
52   (efq       : ∀ {φ}, motive (⊥ ➝ φ) efq)
53   (lem       : ∀ {φ}, motive (φ ⋎ ∼φ) lem)

54
  (mdp       : ∀ {φ ψ}, (hφψ : ⊢ φ ➝ ψ) → (hφ : ⊢ φ) → motive (φ ➝ ψ) hφψ → motive φ hφ 

→ motive ψ (mdp hφψ hφ))
55   : ∀ {φ}, (d : ⊢ φ) → motive φ d := by
56   rintro φ ⟨d⟩; induction d <;> grind;
57
58 @[grind <=]
59 lemma afortiori : ⊢ φ → ⊢ ψ ➝ φ := λ h => mdp impK h
60
61 @[grind .]
62 lemma impId : ⊢ φ ➝ φ := by
63   have h₁ : ⊢ φ ➝ (φ ➝ φ) ➝ φ := impK;
64   have h₂ : ⊢ (φ ➝ (φ ➝ φ) ➝ φ) ➝ (φ ➝ φ ➝ φ) ➝ φ ➝ φ := impS;
65   have h₃ : ⊢ (φ ➝ φ ➝ φ) ➝ φ ➝ φ := mdp h₂ h₁;
66   have h₄ : ⊢ φ ➝ φ ➝ φ := impK;
67   exact mdp h₃ h₄;
68
69 end Provable
70
71
72
73
74 inductive Deduction : Theory α → Formula α → Type*
75 | of_context! {Γ φ} : φ ∈ Γ → Deduction Γ φ
76 | of_proof!  {Γ φ} : ⊢! φ → Deduction Γ φ
77 | mdp! {Γ φ ψ} : Deduction Γ (φ ➝ ψ) → Deduction Γ φ → Deduction Γ ψ
78 infix:40 " ⊢! " => Deduction
79
80 def Deducible (Γ : Theory α) (φ : Formula α) := Nonempty (Γ ⊢! φ)
81 infix:40 " ⊢ " => Deducible
82
83 abbrev Undeducible (Γ : Theory α) (φ : Formula α) := ¬(Γ ⊢ φ)
84 infix:40 " ⊬ " => Undeducible
85
86
87 namespace Deducible
88
89 @[grind ⇐] lemma of_context (h : φ ∈ Γ) : Γ ⊢ φ := ⟨Deduction.of_context! h⟩
90
91 lemma of_proof (h : ⊢! φ) : Γ ⊢ φ := ⟨Deduction.of_proof! h⟩
92
93 @[grind ⇐] lemma of_provable (h : ⊢ φ) : Γ ⊢ φ := of_proof h.some
94

95
@[grind! →] lemma mdp : Γ ⊢ φ ➝ ψ → Γ ⊢ φ → Γ ⊢ ψ := λ ⟨d₁⟩ ⟨d₂⟩ => ⟨Deduction.mdp! d₁ 

d₂⟩

96
97
98 @[grind .] lemma andElimL : Γ ⊢ (φ ⋏ ψ) ➝ φ := by grind;
99 @[grind .] lemma andElimR : Γ ⊢ (φ ⋏ ψ) ➝ ψ := by grind;
100 @[grind .] lemma andIntro : Γ ⊢ φ ➝ ψ ➝ (φ ⋏ ψ) := by grind;
101 @[grind .] lemma orIntroL : Γ ⊢ φ ➝ (φ ⋎ ψ) := by grind;
102 @[grind .] lemma orIntroR : Γ ⊢ ψ ➝ (φ ⋎ ψ) := by grind;
103 @[grind .] lemma orElim : Γ ⊢ (φ ⋎ ψ) ➝ (φ ➝ χ) ➝ (ψ ➝ χ) ➝ χ := by grind;
104 @[grind .] lemma impS : Γ ⊢ (φ ➝ (ψ ➝ χ)) ➝ (φ ➝ ψ) ➝ (φ ➝ χ) := by grind;
105 @[grind .] lemma impK : Γ ⊢ φ ➝ ψ ➝ φ := by grind;
106 @[grind .] lemma efq : Γ ⊢ ⊥ ➝ φ := by grind;
107 @[grind .] lemma lem : Γ ⊢ φ ⋎ ∼φ := by grind;
108
109 @[grind →]
110 lemma ruleAndElimL : Γ ⊢ φ ⋏ ψ → Γ ⊢ φ := by
111   intro h;
112   apply mdp _ h;
113   exact andElimL;
114
115 @[grind →]
116 lemma ruleAndElimR : Γ ⊢ φ ⋏ ψ → Γ ⊢ ψ := by
117   intro h;
118   apply mdp _ h;
119   exact andElimR;
120
121 @[grind →]
122 lemma ruleAndIntro : Γ ⊢ φ → Γ ⊢ ψ → Γ ⊢ φ ⋏ ψ := by
123   intro hφ hψ;
124   apply mdp _ hψ;
125   apply mdp _ hφ;
126   exact andIntro;
127
128 lemma ruleOrIntroL : Γ ⊢ φ → Γ ⊢ φ ⋎ ψ := by
129   intro h;
130   apply mdp _ h;
131   exact orIntroL;
132
133 lemma ruleOrIntroR : Γ ⊢ ψ → Γ ⊢ φ ⋎ ψ := by
134   intro h;
135   apply mdp _ h;
136   exact orIntroR;
137
138 @[grind →]
139 lemma ruleOrElim : Γ ⊢ φ ⋎ ψ → Γ ⊢ φ ➝ χ → Γ ⊢ ψ ➝ χ → Γ ⊢ χ := by
140   intro h₁ h₂ h₃;
141   apply mdp _ h₃;
142   apply mdp _ h₂;
143   apply mdp _ h₁;
144   exact orElim;
145
146 @[grind <=] lemma afortiori : Γ ⊢ φ → Γ ⊢ ψ ➝ φ := λ h => mdp impK h
147

148
@[grind <=] lemma ruleImpS : Γ ⊢ φ ➝ (ψ ➝ χ) → Γ ⊢ φ ➝ ψ → Γ ⊢ φ ➝ χ := λ h₁ h₂ => mdp 

(mdp impS h₁) h₂
149
150 @[grind .] lemma impId : Γ ⊢ φ ➝ φ := of_provable Provable.impId
151
152
153 @[induction_eliminator]
154 lemma rec
155   {motive   : (Γ : Theory α) → (φ : Formula α) → (Γ ⊢ φ) → Sort}
156   (of_context  : ∀ {Γ φ} (h : φ ∈ Γ), motive Γ φ (of_context h))
157   (of_provable : ∀ {Γ φ} (h : ⊢ φ), motive Γ φ (of_provable h))

158
  (mdp         : ∀ {Γ φ ψ}, (hφψ : Γ ⊢ φ ➝ ψ) → (hφ : Γ ⊢ φ) → motive Γ (φ ➝ ψ) hφψ → 

motive Γ φ hφ → motive Γ ψ (mdp hφψ hφ))
159   : ∀ {Γ φ}, (d : Γ ⊢ φ) → motive Γ φ d := by
160   rintro Γ φ ⟨d⟩;
161   induction d with
162   | of_context! h => apply of_context h;

163   | of_proof! h => apply of_provable ⟨h⟩;
164   | mdp! d₁ d₂ ih₁ ih₂ =>
165     apply mdp;
166     . exact ih₁;
167     . exact ih₂;
168
169 lemma rec_provable
170   {motive : (Γ : Theory α) → (φ : Formula α) → (Γ ⊢ φ) → Sort}
171   (of_context : ∀ {Γ φ} (h : φ ∈ Γ), motive Γ φ (of_context h))

172
  (mdp        : ∀ {Γ φ ψ}, (hφψ : Γ ⊢ φ ➝ ψ) → (hφ : Γ ⊢ φ) → motive Γ (φ ➝ ψ) hφψ → 

motive Γ φ hφ → motive Γ ψ (mdp hφψ hφ))
173   (andElimL   : ∀ {Γ φ ψ}, motive Γ ((φ ⋏ ψ) ➝ φ) andElimL)
174   (andElimR   : ∀ {Γ φ ψ}, motive Γ ((φ ⋏ ψ) ➝ ψ) andElimR)
175   (andIntro   : ∀ {Γ φ ψ}, motive Γ (φ ➝ ψ ➝ (φ ⋏ ψ)) andIntro)
176   (orIntroL   : ∀ {Γ φ ψ}, motive Γ (φ ➝ (φ ⋎ ψ)) orIntroL)
177   (orIntroR   : ∀ {Γ φ ψ}, motive Γ (ψ ➝ (φ ⋎ ψ)) orIntroR)
178   (orElim     : ∀ {Γ φ ψ χ}, motive Γ ((φ ⋎ ψ) ➝ (φ ➝ χ) ➝ (ψ ➝ χ) ➝ χ) orElim)
179   (impS       : ∀ {Γ φ ψ χ}, motive Γ ((φ ➝ (ψ ➝ χ)) ➝ (φ ➝ ψ) ➝ (φ ➝ χ)) impS)
180   (impK       : ∀ {Γ φ ψ}, motive Γ (φ ➝ ψ ➝ φ) impK)
181   (efq        : ∀ {Γ φ}, motive Γ (⊥ ➝ φ) efq)
182   (lem        : ∀ {Γ φ}, motive Γ (φ ⋎ ∼φ) lem)
183   : ∀ {φ}, (d : Γ ⊢ φ) → motive Γ φ d := by
184   rintro φ d;
185   induction d using Deducible.rec with
186   | of_provable h => induction h <;> grind;
187   | _ => grind;
188
189 lemma weakening (hΓΔ : Γ ⊆ Δ) (h : Γ ⊢ φ) : Δ ⊢ φ := by
190   induction h using Deducible.rec with
191   | of_context h =>
192     apply of_context;
193     exact hΓΔ h;
194   | of_provable h =>
195     apply of_provable;
196     exact h;
197   | mdp d₁ d₂ ih₁ ih₂ =>
198     apply mdp;
199     . apply ih₁ hΓΔ;
200     . apply ih₂ hΓΔ;
201
202 @[grind <=]
203 lemma weakening_insert (h : Γ ⊢ ψ) : insert φ Γ ⊢ ψ := weakening (by grind) h
204
205 @[grind =]
206 theorem deduction_theorem : (insert φ Γ) ⊢ ψ ↔ Γ ⊢ (φ ➝ ψ) := by
207   constructor;
208   . intro h;
209     generalize eΔ : insert φ Γ = Δ at h;
210     induction h using Deducible.rec with
211     | of_context h =>
212       subst eΔ;
213       simp at h;
214       rcases h with (rfl | h);
215       . exact impId;
216       . apply afortiori
217         apply of_context h;
218     | of_provable h =>
219       apply of_provable;
220       exact Provable.afortiori h;
221     | mdp _ _ ih₁ ih₂ =>
222       subst eΔ;
223       replace ih₁ := ih₁ (by grind);
224       replace ih₂ := ih₂ (by grind);
225       apply ruleImpS ih₁ ih₂;
226   . intro h;
227     apply mdp (weakening_insert h);
228     apply of_context;
229     simp;
230

231 lemma ruleImpTrans : Γ ⊢ φ ➝ ψ → Γ ⊢ ψ ➝ χ → Γ ⊢ φ ➝ χ := by
232   intro h₁ h₂;
233   replace h₁ : insert φ Γ ⊢ ψ := deduction_theorem.mpr h₁;
234   replace h₂ : insert φ Γ ⊢ ψ ➝ χ := weakening_insert h₂;
235   exact deduction_theorem.mp $ mdp h₂ h₁;
236
237 lemma ruleEFQNeg : Γ ⊢ ∼φ → Γ ⊢ φ ➝ ψ := λ h => ruleImpTrans h efq
238
239 lemma ruleDNE (h : Γ ⊢ ∼∼φ) : Γ ⊢ φ := by
240   apply ruleOrElim lem;
241   . show Γ ⊢ φ ➝ φ;
242     exact impId;
243   . apply ruleEFQNeg h;
244
245 end Deducible

1 import Logic101.CPL.Formula
2 import Logic101.CPL.Hilbert
3
4 variable {α : Type u} {φ ψ χ : Formula α} {Γ Δ : Theory α}
5
6 abbrev Valuation (α) := α → Prop
7
8 @[grind]
9 def Forces (V : Valuation α) : Formula α → Prop

10 | #a        => V a
11 | ⊥         => False
12 | φ ⋏ ψ     => Forces V φ ∧ Forces V ψ
13 | φ ⋎ ψ     => Forces V φ ∨ Forces V ψ
14 | φ ➝ ψ     => Forces V φ → Forces V ψ
15 infix:50 " ⊩ " => Forces
16
17 variable {V : Valuation α}
18
19 @[grind .] lemma forces_top : V ⊩ ⊤ := by grind;
20 @[grind =] lemma forces_neg : V ⊩ ∼φ ↔ ¬(V ⊩ φ) := by grind;
21
22
23 @[grind .]
24 def Valid (φ : Formula α) : Prop := ∀ V, V ⊩ φ
25 prefix:50 "⊨ " => Valid
26 alias Tautology := Valid
27
28
29 @[grind .]

30
def ConsequenceOn (V : Valuation α) (Γ : Theory α) (φ : Formula α) : Prop := (∀ ψ ∈ Γ, 

V ⊩ ψ) → V ⊩ φ
31 notation Γ " ⊨[" V "] " φ => ConsequenceOn V Γ φ
32
33 @[grind .]
34 def Consequence (Γ : Theory α) (φ : Formula α) : Prop := ∀ V, Γ ⊨[V] φ
35 infix:50 " ⊨ " => Consequence
36
37 @[grind =]

38
lemma iff_valid_forall_forces : ⊨ φ ↔ ∅ ⊨ φ := by simp [Valid, Consequence, 

ConsequenceOn]
39
40 theorem sound : Γ ⊢ φ → Γ ⊨ φ := by
41   rintro h;
42   induction h using Deducible.rec_provable with
43   | orElim =>
44     intro V hV;
45     grind [Forces];
46   | mdp _ _ ihφψ ihφ =>
47     intro V hV;
48     exact ihφψ V hV $ ihφ V hV;
49   | _ => grind;
50

51 lemma sound_provable : ⊢ φ → ⊨ φ := by
52   intro h;
53   apply iff_valid_forall_forces.mpr;
54   exact sound (Γ := ∅) (Deducible.of_provable h);
55
56 theorem consistent : ⊬ (⊥ : Formula α) := by
57   by_contra h;
58   apply sound_provable h (λ _ => True);
59
60
61
62
63 open Deducible
64
65 namespace Theory
66
67 variable {Γ : Theory α}
68
69 abbrev Inconsistent (Γ : Theory α) : Prop := Γ ⊢ ⊥
70
71 abbrev Consistent (Γ : Theory α) : Prop := ¬(Γ.Inconsistent)
72

73
@[grind =] lemma iff_inconsistent_insert : (insert φ Γ).Inconsistent ↔ Γ ⊢ ∼φ := 

deduction_theorem
74

75
@[grind =] lemma iff_consistent_insert : (insert φ Γ).Consistent ↔ Γ ⊬ ∼φ := 

deduction_theorem.not
76
77 @[grind =>]

78
lemma either_consistent_insert_of_consistent (Γ_consis : Γ.Consistent) (φ) : (insert φ 

Γ).Consistent ∨ (insert (∼φ) Γ).Consistent := by
79   contrapose! Γ_consis;
80   obtain ⟨h₁, h₂⟩ := Γ_consis;
81   apply ruleOrElim lem;
82   . show Γ ⊢ φ ➝ ⊥;
83     grind;
84   . grind
85
86 @[grind =>]

87
lemma either_unprovable_of_consistent (Γ_consis : Γ.Consistent) (φ) : Γ ⊬ φ ∨ Γ ⊬ ∼φ := 

by
88   contrapose! Γ_consis;
89   obtain ⟨hφ, hnφ⟩ := Γ_consis;
90   apply mdp hnφ hφ;
91
92 @[grind →]

93
lemma unprovable_neg_of_provable_of_consistent (Γ_consis : Γ.Consistent) (h : Γ ⊢ φ) : 

Γ ⊬ ∼φ := by grind;
94
95 @[grind →]

96
lemma unprovable_of_provable_neg_of_consistent (Γ_consis : Γ.Consistent) (h : Γ ⊢ ∼φ) : 

Γ ⊬ φ := by grind;
97
98 @[grind →]
99 lemma not_mem_bot_of_consistent (Γ_consis : Γ.Consistent) : ⊥ ∉ Γ := by
100   by_contra! h;
101   apply Γ_consis;
102   exact of_context h;
103
104 @[grind →]

105
lemma consistent_insert_neg_of_unprovable (h : Γ ⊬ φ) : (insert (∼φ) Γ).Consistent := 

by
106   contrapose! h;
107   apply ruleDNE;
108   apply iff_inconsistent_insert.mp h;
109
110 abbrev Maximal (Γ : Theory α) : Prop := ∀ φ, φ ∈ Γ ∨ ∼φ ∈ Γ

111
112 end Theory
113
114
115 abbrev MaximalConsistentTheory (α) := { Γ : Theory α // Γ.Consistent ∧ Γ.Maximal }
116
117
118 namespace Theory
119
120 open Encodable
121
122 open Classical in

123
def lindenbaum_expand (Γ : Theory α) (φ : Formula α) : Theory α := if Theory.Consistent 

(insert φ Γ) then insert φ Γ else Γ
124
125 @[grind =>]

126
lemma consistent_lindenbaum_expand (Γ_consis : Theory.Consistent Γ) : Theory.Consistent 

(lindenbaum_expand Γ φ) := by
127   dsimp [lindenbaum_expand];
128   split <;> assumption;
129
130 variable [Encodable α]
131
132 def lindenbaum_indexed (Γ : Theory α) : ℕ → Theory α
133   | 0     => Γ
134   | n + 1 =>
135     match (decode n) with
136     | some φ => lindenbaum_expand (lindenbaum_indexed Γ n) φ
137     | none   => lindenbaum_indexed Γ n
138
139 variable {Γ : Theory α} {n : ℕ}
140
141 @[grind =>]

142
lemma consistent_lindenbaum_indexed (Γ_consis : Theory.Consistent Γ) : 

Theory.Consistent (lindenbaum_indexed Γ n) := by
143   induction n with
144   | zero => assumption;
145   | succ n ih =>
146     dsimp [lindenbaum_indexed];
147     split;
148     . grind [consistent_lindenbaum_expand];
149     . assumption;
150

151
lemma monotonic_lindenbaum_indexed (h : n ≤ m) : lindenbaum_indexed Γ n ⊆ 

lindenbaum_indexed Γ m := by
152   induction h with
153   | refl => simp;
154   | step h ih =>
155     trans;
156     . exact ih;
157     . dsimp [lindenbaum_indexed, lindenbaum_expand];
158       split;
159       . split <;> grind;
160       . grind;
161
162 def lindenbaum_maximal (Γ : Theory α) : Theory α := ⋃ n, lindenbaum_indexed Γ n
163

164
lemma iff_mem_lindenbaum_maximal : φ ∈ Γ.lindenbaum_maximal ↔ ∃ n, φ ∈ 

Γ.lindenbaum_indexed n := by
165   simp [lindenbaum_maximal, Set.mem_iUnion]
166
167 @[grind .]

168
lemma subset_lindenbaum_indexed_lindenbaum_maximal : lindenbaum_indexed Γ n ⊆ 

lindenbaum_maximal Γ := by
169   intro φ hφ;
170   apply iff_mem_lindenbaum_maximal.mpr;
171   use n;

172

173
lemma subset_lindenbaum_maximal : Γ ⊆ lindenbaum_maximal Γ := 

subset_lindenbaum_indexed_lindenbaum_maximal (n := 0)
174

175
lemma exists_deducible_maximal (h : Γ.lindenbaum_maximal ⊢ φ) : ∃ n, lindenbaum_indexed 

Γ n ⊢ φ := by
176   generalize eΔ : Γ.lindenbaum_maximal = Δ at h;
177   induction h using Deducible.rec with
178   | of_context h =>
179     subst eΔ;
180     obtain ⟨n, hn⟩ := iff_mem_lindenbaum_maximal.mp h;
181     use n;
182     apply of_context hn;
183   | of_provable h =>
184     use 0;
185     exact of_provable h;
186   | mdp d₁ d₂ ih₁ ih₂ =>
187     obtain ⟨n, ih₁⟩ := ih₁ eΔ;
188     obtain ⟨m, ih₂⟩ := ih₂ eΔ;
189     use max n m;
190     apply mdp;
191     . exact weakening (monotonic_lindenbaum_indexed (by omega)) ih₁;
192     . exact weakening (monotonic_lindenbaum_indexed (by omega)) ih₂;
193
194 @[grind .]

195
lemma consistent_lindenbaum_maximal (Γ_consis : Theory.Consistent Γ) : 

Theory.Consistent (Γ.lindenbaum_maximal) := by
196   by_contra hC;
197   obtain ⟨_, hn⟩ := exists_deducible_maximal hC;
198   apply consistent_lindenbaum_indexed Γ_consis hn;
199

200
lemma maximal_lindenbaum_maximal (Γ_consis : Theory.Consistent Γ) : Theory.Maximal 

(Γ.lindenbaum_maximal) := by
201   intro φ;

202
  rcases Γ.lindenbaum_maximal.either_unprovable_of_consistent 

(consistent_lindenbaum_maximal Γ_consis) φ with (hC | hC);
203   . right;
204     apply iff_mem_lindenbaum_maximal.mpr;
205     use ((encode (∼φ)) + 1);
206     suffices (insert (∼φ) (Γ.lindenbaum_indexed (encode (∼φ)))).Consistent by
207       simp [lindenbaum_indexed, lindenbaum_expand, this];
208     apply iff_consistent_insert.mpr;
209     contrapose! hC;
210     apply weakening (by grind) (ruleDNE hC);
211   . left;
212     apply iff_mem_lindenbaum_maximal.mpr;
213     use ((encode φ) + 1);
214     suffices (insert φ (Γ.lindenbaum_indexed (encode φ))).Consistent by
215       simp [lindenbaum_indexed, lindenbaum_expand, this]
216     apply iff_consistent_insert.mpr;
217     contrapose! hC;
218     apply weakening (by grind) hC;
219

220
lemma lindenbaum (Γ : Theory α) (Γ_consis : Theory.Consistent Γ) : ∃ Δ : 

MaximalConsistentTheory α, Γ ⊆ Δ.1 := by

221
  use ⟨Γ.lindenbaum_maximal, consistent_lindenbaum_maximal Γ_consis, 

maximal_lindenbaum_maximal Γ_consis⟩;
222   exact subset_lindenbaum_maximal;
223
224 end Theory
225
226
227 namespace MaximalConsistentTheory
228
229 variable {Γ : MaximalConsistentTheory α}
230
231 lemma consistent (Γ : MaximalConsistentTheory α) : Γ.1.Consistent := Γ.2.1

232
233 @[grind .] lemma maximal : Theory.Maximal Γ.1 := Γ.2.2
234
235 @[grind =]
236 lemma iff_mem_provable : φ ∈ Γ.1 ↔ Γ.1 ⊢ φ := by
237   constructor;
238   . apply of_context;
239   . intro h;
240     suffices ∼φ ∉ Γ.1 by grind;
241     by_contra hC;
242     apply Γ.consistent;
243     apply mdp;
244     . show Γ.1 ⊢ ∼φ;
245       apply of_context hC;
246     . exact h;
247
248 @[simp, grind .] lemma not_mem_bot : ⊥ ∉ Γ.1 := by grind;
249
250 @[grind =] lemma iff_mem_neg : ∼φ ∈ Γ.1 ↔ φ ∉ Γ.1 := by constructor <;> grind;
251
252 @[grind =] lemma iff_not_mem_provable : φ ∉ Γ.1 ↔ Γ.1 ⊢ ∼φ := by grind;
253
254 @[grind =]
255 lemma iff_mem_imp : (φ ➝ ψ) ∈ Γ.1 ↔ (φ ∈ Γ.1 → ψ ∈ Γ.1) := by
256   constructor;
257   . simp only [iff_mem_provable]
258     apply mdp;
259   . rintro h;
260     rcases not_or_of_imp h with (hφ | hψ);
261     . replace hφ := iff_mem_neg.mpr hφ;
262       simp only [iff_mem_provable] at hφ ⊢;
263       exact ruleEFQNeg hφ;
264     . simp only [iff_mem_provable] at hψ ⊢;
265       exact afortiori hψ;
266
267 @[grind =]
268 lemma iff_mem_and : (φ ⋏ ψ) ∈ Γ.1 ↔ φ ∈ Γ.1 ∧ ψ ∈ Γ.1 := by
269   simp only [iff_mem_provable]
270   constructor;
271   . intro h;
272     constructor;
273     . apply ruleAndElimL h;
274     . apply ruleAndElimR h;
275   . simp only [and_imp];
276     apply ruleAndIntro;
277
278 @[grind =]
279 lemma iff_mem_or : (φ ⋎ ψ) ∈ Γ.1 ↔ φ ∈ Γ.1 ∨ ψ ∈ Γ.1 := by
280   constructor;
281   . intro h;
282     by_contra! hC;
283     obtain ⟨hφ, hψ⟩ := hC;
284     replace hφ : ↑Γ ⊢ ∼φ := iff_not_mem_provable.mp hφ;
285     replace hψ : ↑Γ ⊢ ∼ψ := iff_not_mem_provable.mp hψ;
286     apply Γ.consistent;
287     apply ruleOrElim (iff_mem_provable.mp h) hφ hψ;
288   . simp only [iff_mem_provable];
289     rintro (hφ | hψ);
290     . apply ruleOrIntroL;
291       exact hφ;
292     . apply ruleOrIntroR;
293       exact hψ;
294
295 end MaximalConsistentTheory
296
297 theorem complete [Encodable α] : Γ ⊨ φ → Γ ⊢ φ := by
298   contrapose!;
299   intro h;
300   suffices ∃ V, (∀ ψ ∈ Γ, V ⊩ ψ) ∧ ¬V ⊩ φ by simpa [Consequence, ConsequenceOn];
301

302
  obtain ⟨Δ, hΓΔ⟩ := (insert (∼φ) Γ).lindenbaum $ by grind 

[Theory.consistent_insert_neg_of_unprovable];
303
304   let V : Valuation α := λ a => (#a ∈ Δ.1);
305   have hV : ∀ {ψ}, V ⊩ ψ ↔ ψ ∈ Δ.1 := by
306     intro ψ;
307     induction ψ <;> grind
308   use V;
309   constructor;
310   . intro γ hγ;
311     apply hV.mpr;
312     apply hΓΔ;
313     grind;
314   . apply hV.not.mpr;
315     apply Δ.iff_mem_neg.mp;
316     apply hΓΔ;
317     grind;

